
Bei der Überprüfung von Datumsangaben beispielsweise in einem Formular kommen häufig Abfragen in der folgenden Form zum Einsatz:

If Not IsDate(Me.txtDatumVon) Then
  MsgBox "Ungültiges Datum!"
  Me.txtDatumVon.SetFocus
End If

Dabei besteht die Gefahr, dass falsche Datumsangaben gespeichert werden, denn Access bzw. VBA kennt nur den Datentyp "Date", der sowohl Datums-, wie auch Zeitangaben ("12.3.2006", "17:23:45") oder kombinierte Datums-/Zeitangaben ("12.3.2006 17:23:45") aufnimmt. Eine Abfrage wie

IsDate(Me.txtDatumVon)

liefert somit für Eingaben wie "12.3.2006" oder "12.3.2006 17:23:45" das Ergebnis "True". Sie liefert aber auch bei Eingaben wie "17:23:45" das Ergebnis "True" - und hier lauert die Gefahr der Speicherung falscher Datumsangaben. Beispiel: Sie verwalten in einer Datenbank Tätigkeitsnachweise, dazu gibt es unter anderem vier Felder "Datum von", "Datum bis", "Zeit von" und "Zeit bis". Wenn bei der Eingabe nun beispielsweise die Felder "Datum von" und "Zeit von" verwechselt werden, also im Feld "Datum von" eine Uhrzeit wie "8:00" eingegeben wird, liefert die Abfrage "IsDate(Me.txtDatumVom)" das Ergebnis "True", denn "8:00" ist für ein Feld vom Datentyp "Date" ein gültiger Wert.

Das Problem: Ist nun für das Feld der Datentyp "Date" zum Beispiel mit dem Format "Datum, kurz" eingestellt, so konvertiert Access die gemäß "IsDate()" gültige Eingabe "8:00" in ein Datum - was dabei herauskommt, können Sie wie folgt nachvollziehen:

1. Legen Sie eine Tabelle "DZTest" mit einem Feld "DatumVon" vom Datentyp "Datum/Zeit" mit dem Format "Datum, kurz" an.

2. Lassen Sie die Tabelle in der Datenblattansicht anzeigen.

3. Geben Sie im Feld "DatumVon" eine Uhrzeit wie "8:00" ein.

Sobald Sie die Return-Taste drücken, konvertiert Access die Eingabe, es wird jetzt "30.12.1899" angezeigt. Soll nun beispielsweise das Honorar auf Tagesbasis mit einem Satz von 250 Euro/Tag für eine Tätigkeit im Zeitraum "Datum von" bis "Datum bis" berechnet werden, ergibt eine Formel wie die folgende astronomische Beträge:

dblHonorar = DateDiff("d", DatumBis, DatumVon) * 250

Ist "Datum bis" zum Beispiel "4.5.2006", ergibt sich eine Differenz zum "30.12.1899" von 38.836 Tagen und somit ein Honorar von 9.709.000 Euro.

Was können Sie tun, um eine falsche Speicherung von Datums-Angaben zu verhindern? Der sicherste Weg ist die Verwendung von Eingabeformaten: Damit legen Sie explizit fest, dass beispielsweise in einem Feld "Datum von" nur ein reines Datum oder in einem Feld "Zeit bis" nur eine reine Zeitangabe eingegeben werden kann:

1. Öffnen Sie die betreffende Tabelle im Entwurfsmodus.

2. Markieren Sie der Reihe nach die Felder, die ein reines Datum aufnehmen sollen.

3. Stellen Sie "Format" auf "Datum, kurz" ein.

4. Klicken Sie ins Feld "Eingabeformat" und dann auf die Schaltfläche mit den drei Punkten.

5. Markieren Sie im Eingabeformat-Assistenten den Eintrag "Datum, kurz". Achtung! Access 97 verwendet bei "Datum, kurz" für die Eingabemaske eine zweistellige Jahreszahl! Klicken Sie auf die Schaltfläche Weiter und ergänzen Sie auf der folgenden Seite des Assistenten die Vorgabe "99/99/00" im Feld Eingabeformat um zwei weitere Nullen!

6. Klicken Sie auf die Schaltfläche Fertigstellen.

7. Markieren Sie der Reihe nach die Felder, die eine reine Zeitangabe aufnehmen sollen.

8. Stellen Sie "Format" je nach Erfordernis auf "Zeit, 12 Std." oder "Zeit, 24 Std." ein.

9. Klicken Sie ins Feld "Eingabeformat" und dann auf die Schaltfläche mit den drei Punkten.

10. Markieren Sie im Eingabeformat-Assistenten den zur Einstellung "Format" korrespondieren Eintrag, also "Zeit, 12 Std." oder "Zeit, 24 Std.".

11. Klicken Sie auf die Schaltfläche Fertigstellen.

12. Speichern Sie die Änderungen am Tabellenentwurf.

13. Wenn basierend auf der jeweiligen Tabelle bereits Formulare angelegt sind, wiederholen Sie die oben erläuterten Einstellungen bezüglich "Format" und "Eingabeformat" (nur Formulare) für die betreffenden Formulare.

In Zukunft trägt Access nun dafür Sorge, dass in ein Feld wie "Datum von" auch wirklich nur eine Datumsangabe und nicht aus Versehen eine Zeitangabe eingegeben werden kann und zeigt nötigenfalls eine Fehlermeldung an.

Wie Sie Datums- und Zeitangaben zuverlässig in VBA unterscheiden und prüfen, ist Inhalt des zweiten Teils dieses Tipps in der nächsten Ausgabe.

In der letzten Ausgabe hatten wir Ihnen gezeigt, wie Sie Eingaben von Datums- und Zeitangaben über Eingabeformate absichern. Das schützt jedoch nicht vor falschen Datums- oder Zeitangaben, die zum Beispiel durch einen Import oder eine Anfügeabfrage in eine Tabelle geraten. Sie müssen also in jedem Fall Datums- und Zeitangaben vor ihrer Weiterverarbeitung per VBA zuverlässig unterscheiden und prüfen können. Sie benötigen dazu drei benutzerdefinierte Funktionen: "IsDateAndTime()" für die Überprüfung auf "Kombiwerte" sowie "IsDateOnly()" und "IsTimeOnly()" um festzustellen, ob reine Datums- oder Zeitangaben vorliegen. Kopieren Sie zunächst die folgenden Funktionen in ein neues oder in ein vorhandenes Modul Ihrer Datenbank:

Function IsDateAndTime(varValue) As Boolean
  Dim I As Long, L As Long

  On Error Resume Next
  IsDateAndTime = False
  If IsNull(varValue) Then 'Feld ist leer
     Exit Function
  ElseIf Not IsDate(varValue) Then
    'Ist weder Datum, Datum/Zeit noch Zeit
    Exit Function
  ElseIf IsTimeOnly(varValue) Then 'Nur Zeitangabe
    Exit Function
  ElseIf InStr(varValue, ":") = 0 Then
    'Keine Zeit enthalten
    Exit Function
  End If
  IsDateAndTime = True

End Function

Hier wird zunächst mit "IsNull()" und "IsDate()" geprüft, ob ein Inhalt vorliegt und ob er als Datum, Datum/Zeit oder Zeit formatiert ist. Ist das nicht der Fall, wird die Funktion verlassen und das Default-Ergebnis "False" gemeldet. Andernfalls prüfen wir mit der benutzerdefinierten Funktion "IsTimeOnly()" (siehe unten), ob es sich um eine reine Zeitangabe handelt. Wenn ja, wird die Funktion ebenfalls mit Rückmeldung "False" verlassen. Eine letzte Abfrage prüft dann anhand des Vorhandenseins eines Doppelpunktes als Trennzeichen für Zeitangaben, ob eine Uhrzeit enthalten ist. Trifft das nicht zu, wird die Funktion auch mit Rückmeldung "False" verlassen.

Übrig bleibt dann nur noch die Möglichkeit, dass es sich um eine Kombination aus Datum und Zeit handelt, so dass als Funktionsergebnis "True" gesetzt werden kann. Wenn Sie nun zum Beispiel bei einer Weiterverarbeitung per VBA davon ausgehen, dass ein Standarddatum, also eine Kombination aus Datum/Zeit wie zum Beispiel "12.03.2008 13:45:27" vorliegt, nehmen Sie eine entsprechende Prüfung wie folgt vor:

If IsDateAndTime(Feld) Then
  MyDate = DateValue(Feld)
  MyTime = TimeValue(Feld)
  'Weitere Verarbeitung...
Else 
  MsgBox "Keine gültige Datums-/Zeitangabe!"
End If

Hier reicht also der Aufruf von "IsDateAndTime()", um auf das erwartete Format zu prüfen.

Function IsDateOnly(varValue) As Boolean
  Dim I As Long, L As Long

  On Error Resume Next
  IsDateOnly = False 'Default
  If IsNull(varValue) Then 'Feld ist leer
     Exit Function
  ElseIf Not IsDate(varValue) Then
   'Ist weder Datum, Datum/Zeit noch Zeit
    Exit Function
  ElseIf InStr(varValue, ":") <> 0 Then
      Exit Function
  End If
  IsDateOnly = True

End Function

Hier wird zunächst per "IsNull()" und "IsDate()" geprüft, ob der übergebene Ausdruck "NULL", also das Feld leer ist oder ob er überhaupt eine gültige Datums-, Datums/Zeit- oder Zeitangabe darstellt. Trifft das zu bzw. nicht zu, wird die Funktion verlassen und das zuvor als Default gesetzte Funktionsergebnis "False" gemeldet. Andernfalls prüft die nächste Abfrage, ob im Ausdruck ein Doppelpunkt, also das Trennzeichen für Zeitangaben vorhanden ist. Ist das der Fall, handelt es sich um eine Datums-/Zeitangabe und die Funktion wird ebenfalls mit Rückmeldung "False" verlassen. Übrig bleibt dann nur noch die Möglichkeit, dass es sich um eine reine Datumsangabe handelt, so dass als Funktionsergebnis "True" gesetzt werden kann. In diesem Fall kann also eine direkte Zuweisung erfolgen.

Nehmen wir ein konkretes Beispiel: Soll eine Projektdauer in Tagen per VBA berechnen werden, benötigen Sie nur Datumsangaben. Da eine Datumsangabe sowohl aus einem Feldinhalt "Nur Datum" als auch aus einem Feldinhalt "Datum/Zeit" extrahiert werden kann, müssen Sie wie folgt prüfen:

MyDate = vbNull
If IsDateOnly(Feld) Then
  MyDate = Feld
ElseIf IsDateAndTime(Feld) Then
  MyDate = DateValue(Feld)
Else 
  MsgBox "Kein gültiges Datum!"
End If
If Not IsNull(MyDate) Then
  'Weitere Verarbeitung...
End If

"IsDateOnly()" prüft, ob der übergebene Parameter ein reines Datum darstellt und liefert "True", wenn das zutrifft oder "False", wenn es sich zum Beispiel um eine reine Zeitangabe oder eine Datums-/Zeit-Angabe handelt.

Public Function IsTimeOnly(varTime As Variant) As Boolean

  On Error Resume Next
  IsTimeOnly = False
  If IsNull(varTime) Then
    Exit Function
  ElseIf Not IsDate(varTime) Then
    Exit Function
  Else
    If CStr(varTime) Like "*#*.*#*" = False Then
      If Fix(CDate(varTime)) = 0 Then
        If CStr(varTime) _
           Like "1[3-9]*[aApP]*" = False Then
          If CStr(varTime) Like "2[0-3]*[aApP]*" = _
             False Then IsTimeOnly = True
        End If
      End If
    End If
  End If

End Function

Im Gegensatz zu "IsDateAndTime()" und "IsDateOnly()", bei denen wir sozusagen durch ein Ausschlussverfahren geprüft haben, geht es hier etwas komplizierter zu: Es wird zunächst wieder mit "IsNull()" und "IsDate()" geprüft, ob ein Inhalt vorliegt und ob dieser in einem der Formate Datum, Datum/Zeit oder Zeit vorliegt. Wenn nicht, wird die Funktion verlassen und das Default-Ergebnis "False" gemeldet. Andernfalls prüfen wir mit "CStr(varTime) Like "*#*.*#*"" und "Fix(CDate(varTime)) = 0", ob eine Datumsangabe vorhanden ist und testen mit "CStr(varTime) Like "1[3-9]*[aApP]*"" und "CStr(varTime) Like "2[0-3]*[aApP]*"", ob eine gültige Zeitangabe im 12- oder 24-Stundenformat und ggf. mit Zusatz "AM/PM" oder "am/pm" vorliegt. Erst dann wird das Funktionsergebnis auf "True" gesetzt.

Zum Beispiel: Wenn Sie Feldinhalte für die Berechnung von Stunden bei einem Tätigkeitsnachweis auslesen, sind nur Zeitangaben zu verarbeiten. Diese können Sie aus Feldinhalten, die reine Zeitangaben enthalten oder aus Feldinhalten, die Datums-/Zeitangaben enthalten, gewinnen:

MyTime = vbNull
If IsDateAndTime(Feld) Then
  MyTime = TimeValue(Feld)
ElseIf IsTimeOnly(Feld) Then
  MyTime = Feld
Else 
  MsgBox "Keine gültige Zeitangabe!"
End If
If Not IsNull(MyTime) Then
  'Weitere Verarbeitung...
End If

Hier wird mit "IsDateAndTime()" geprüft, ob eine "Kombiangabe" aus Datum und Zeit vorliegt. Ist das der Fall, kann daraus die Zeitangabe mit Hilfe der VBA-Funktion "TimeValue()" extrahiert werden. Ansonsten prüft "IsTimeOnly()", ob es sich um eine reine Zeitangabe handelt.

